

Exporting OLS Regression Results in Stata

Marco Rosso

Exporting Regression Results in Stata

In empirical work, regression output is rarely ready for publication.

Stata's package `estout` (Jann, 2005) provides a flexible system to produce:

- publication-ready \LaTeX tables,
- compact output for papers, reports, and theses,
- consistent formatting across many model specifications.

We use the workflow:

`eststo` \Rightarrow `estout` (or `esttab`)

This allows storing multiple models and exporting them as a single formatted table.

Workflow Overview

1. Run one or more regressions
2. Store each model with `eststo`
3. Export them together using `estout`

→ `estout` gives full control over:

- variable order and labels,
- significance stars,
- custom column headers,
- fixed-effects indicators,
- spacing and \LaTeX commands.

Storing Models with eststo

Syntax:

```
eststo name: regression_command
```

Example:

```
eststo clear
```

```
eststo m1: reg y x1 x2
```

```
eststo m2: reg y x1 x2 x3
```

```
eststo m3: reg y x1 x2 x3 x4
```

You can now export all models together in a single table.

Why estout Instead of esttab?

esttab:

- easy and fast,
- limited formatting options.

estout:

- full control,
- can build journal-style tables,
- custom multi-column headers,
- custom line spacing,
- fixed-effects indicators,
- total control over \LaTeX .

→ **Use estout for published or thesis-quality tables.**

Anatomy of an estout Command

A full estout table typically has:

1. **Model selection:** choose which stored models to display
2. **Variable selection:** `keep()`, `drop()`, `order()`
3. **Variable labels:** `varlabels()`
4. **Cell formatting:** coefficients, SEs, stars
5. **Custom table structure:**
 - ▶ `prehead()` — lines before coefficient rows
 - ▶ `posthead()` — after header row
 - ▶ `prefoot()` — before final stats
 - ▶ `postfoot()` — closing the table

This allows you to replicate journal-style formatting exactly.

Full Example: Setup

We estimate three OLS models and store them:

```
eststo clear
```

```
eststo m1: reg y x1 x2
```

```
eststo m2: reg y x1 x2 x3
```

```
eststo m3: reg y x1 x2 x3 x4
```

Next, we export them to \LaTeX using `estout`.

Full estout Example (1)

Key idea: Build a journal-quality table entirely from Stata.

```
estout m1 m2 m3 using "table_ols.tex", replace ///
    style(tex) collabels(none) ///
    keep(x1 x2 x3 x4) ///
    order(x1 x2 x3 x4) ///
    varlabels( ///
        x1 "Main regressor 1" ///
        x2 "Main regressor 2" ///
        x3 "Control 1" ///
        x4 "Control 2", ///
        end("\addlinespace[0.15cm] ") ///
    ) ///
    cells(b(fmt(3) star) se(fmt(3) par)) ///
    starlevels(* 0.10 ** 0.05 *** 0.01) ///
```


Full estout Example (2)

Adding statistics and fixed-effects indicators

```
stats(N r2, ///  
      fmt(%9.0g 3) ///  
      labels("Observations" "R-squared") ///  
      layout(@ "\multicolumn{1}{c}{@}") ///  
) ///  
indicate("Year FE = yearFE" "Region FE = regionFE", ///  
         labels("\multicolumn{1}{c}{\checkmark}" ///  
               "\multicolumn{1}{c}{ }") ///  
) ///
```

Full estout Example (3)

Adding L^AT_EX structure (header, footer, spacing)

```
prehead( ///  
  "\begin{tabular}{l c c c}" ///  
  "\hline\hline" ///  
  "\addlinespace[0.1cm]" ///  
  " & \multicolumn{3}{c}{Dependent variable: Y}" ///  
  "\cmidrule(lr){2-4}" ///  
  " & (1) & (2) & (3)" ///  
) ///  
posthead("\addlinespace[0.1cm] \hline \addlinespace[0.05cm]") ///  
prefoot("\addlinespace[0.05cm] \hline \addlinespace[0.05cm]") ///  
postfoot( ///  
  "\addlinespace[0.1cm] \hline\hline" ///  
  "\end{tabular}" ///  
) ///  
end("\[-0.10cm]")
```

What This Table Produces

The previous code produces a table with:

- clean top and bottom rules (`\hline\hline`),
- custom spacing via `\addlinespace`,
- multi-column header for the dependent variable,
- custom column labels (1), (2), (3),
- labelled regressors and grouped controls,
- standard errors in parentheses,
- significance stars,
- fixed-effect indicators,
- observations and R-squared rows.

This is the standard journal style used in empirical economics.

Key Takeaways

- ✓ `eststo` stores each model cleanly.
- ✓ `estout` allows full control of \LaTeX formatting.
- ✓ You can fully replicate published-table style directly from Stata.
- ✓ Always control: variable order, labels, headers, fixed effects, spacing.
- ✓ Your tables become fully reproducible and consistent across drafts.